Royal St. George’s College		Design Engineering Studio
Shiftout DAC with Feedback
Advanced Computer Engineering School	TEJ3M - Fundamentals
// PROJECT  :ShiftoutDACwithFeedback
// PURPOSE  :595 shift register feeding an 8-bit R/2R ladder
//           with ADC feedback
// COURSE   :TEJ3M
// AUTHOR   :C. D'Arcy
// DATE     :Created: 2025 10 17.
// MCU      :328p (Nano)
// STATUS   :Working
// REFERENCE:Using the Serial Plotter Tool:
//https://docs.arduino.cc/software/ide-v2/tutorials/ide-v2-serial-plotter/

#define DATA 11   //aka MOSI
#define CLOCK 8   //any pin will do (for now)
#define LATCH 10  //aka SS (Slave Select)
#define PACE 30   //speed of animation
#define DACOUT A5 //read the output of the DAC R/2R Ladder
#define VCC 5.0   //supply voltage
uint16_t count;   //ramp up the digital input to the DAC
uint16_t reading; //raw 10-bit ADC reading from the DAC
float voltage;    //map the raw ADC reading to the corresponding voltage 

void setup() {
  Serial.begin(9600);       //invoke the services of the Serial Plotter
  while (!Serial) {}
  pinMode(DATA, OUTPUT);    //prepare the three 595 inputs
  pinMode(LATCH, OUTPUT);
  pinMode(CLOCK, OUTPUT);
  count = 0;                //start the count off at ground
}

void loop() {
  digitalWrite(LATCH, LOW);               //disable output registers
  shiftOut(DATA, CLOCK, MSBFIRST, count); //shiftout the 8-bits  
  digitalWrite(LATCH, HIGH);              //present to output pins
  
  reading = analogRead(DACOUT);           //read DAC output(ADC)
        
  voltage = reading*VCC/1024;             //convert to (fp) voltage 
  
  Serial.print("VCC:");                   //publish to the Serial Plotter
  Serial.print(VCC);
  Serial.print(',');
  Serial.print("GND:");
  Serial.print(0.0);
  Serial.print(',');
[bookmark: _GoBack]  Serial.print("ADC_Voltage:");
  Serial.println(voltage);
  
  delay(PACE);                            //pause...
  count++;                                //continue, with rollover
}

